
	Reason for change:
	During the preamble of some Inter-RAT test cases using function f_EUTRA38_NR5GC_PreambleAndSwitchOff(),there is a trigger sent from EUTRA PTC to IMS PTC to indicate that the DEATCH has been done and tell IMS PTC to stop waiting for Re-register. For test cases requiring IMS this trigger is handled inside function f_IMS_IpcanReleaseWithOptionalDeregistration() but for test cases using the IMS default behaviour it is currently not handled and remains in the message queue at IMS PTC “OtherIPCAN” port.

With the MCC implementation of R5s221193, following the preamble EUTRA PTC will send a reset command to IMS (from function f_EUTRA_SwitchOnAndResetIMS()) and will now wait for a CNF for this, but this message gets blocked behind the pending trigger in the queue at the IMS PTC and so is never processed. Due to this the EUTRA PTC now gets blocked and cannot handle the RRCConnectionRequest when it comes from the UE.


	
	

	Summary of change:
	Absorb any trigger coming from EUTRA PTC (OtherIPCan port) in the IMS default handler, so that a following reset message can be processed and responded.

	
	

	Consequences if not approved:
	Some NR5GC IRAT test cases will not execute correctly

1. [bookmark: _Toc122434485][bookmark: _Toc126242285]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D23wk12 related to the title of this CR.
Contact:	Shaun Harry 
	shaun.harry@keysight.com
1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc126242286]Corrections required
1.1 [bookmark: _Toc122605392][bookmark: _Toc126242287][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Change 1
	Function name
	f_IMS_DefaultHandler()

	Reason for change
	During the preamble of some Inter-RAT test cases using function f_EUTRA38_NR5GC_PreambleAndSwitchOff(),there is a trigger sent from EUTRA PTC to IMS PTC to indicate that the DEATCH has been done and tell IMS PTC to stop waiting for Re-register. For test cases requiring IMS this trigger is handled inside function f_IMS_IpcanReleaseWithOptionalDeregistration() but for test cases using the IMS default behaviour it is currently not handled and remaings in the message queue at IMS PTC “OtherIPCAN” port.

With the MCC implementation of R5s221193, following the preamble EUTRA PTC will send a reset command to IMS (from function f_EUTRA_SwitchOnAndResetIMS()) and will now wait for a CNF for this, but this message gets blocked behind the pending trigger in the queue at the IMS PTC and so is never processed. Due to this the EUTRA PTC now gets blocked and cannot handle the RRCConnectionRequest when it comes from the UE.

	Summary of change
	Handle triggers coming from EUTRA PTC (OtherIPCan port) in the IMS default handler, so that a following reset message can be processed and responded.

	TTCN module
	IMS_DefaultHnalder.ttcn

	MCC160 Comment
	



Before Change:
	  
  function f_IMS_DefaultHandler() runs on IMS_PTC
  {
    var IMS_DefaultHandlerInfo_Type v_DefaultHandlerInfo;       /* @sic R5-163201 sic@ */
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_Coordination_MSG;
    var integer v_NonceCountForNextRegister;

    f_IMS_PTC_Init(-, defaultHandling, IMS_Security);
    f_IMS_PTC_CheckAudioVideoTags_Set(false);                   /* @sic R5-201110: set check of audio/video feature tag to be false sic@ */

    v_DefaultHandlerInfo.State := WaitForRegistration;
    v_DefaultHandlerInfo.IpcanType := IPCAN_E;                  /* @sic R5s160133 sic@ */

    while (true) {
      alt {
        [] IPCAN.receive(cmr_IPCAN_IMS_Control) -> value v_IMS_IPCAN_Coordination_MSG {         /* @sic R5s160133 sic@ */
          v_DefaultHandlerInfo := fl_IMS_DefaultHandler_ChangeInfo(v_DefaultHandlerInfo, IPCAN_E, v_IMS_IPCAN_Coordination_MSG.IPCAN_IMS_Control);
        }
        [] OtherIPCAN.receive(cmr_IPCAN_IMS_Control) -> value v_IMS_IPCAN_Coordination_MSG {    /* @sic R5s160133 sic@ */
          v_DefaultHandlerInfo := fl_IMS_DefaultHandler_ChangeInfo(v_DefaultHandlerInfo, OtherIPCAN_E, v_IMS_IPCAN_Coordination_MSG.IPCAN_IMS_Control);
        }
        [] IMS_Server.receive(car_IMS_Register_Request(?)) -> value v_IMS_DATA_REQ {      // receive any REGISTER
          select (v_DefaultHandlerInfo.State) {
            case (WaitForRegistration) {
              f_IMS_DefaultHandler_InitialRegistration(v_DefaultHandlerInfo.IpcanType, v_IMS_DATA_REQ);  // @sic R5s140788 change 2 sic@
              v_DefaultHandlerInfo.State := WaitForDeregistration;
              v_NonceCountForNextRegister := 2;  /* next (re-)registration shall have nounce count of 2 @sic R5-163201 sic@ */
            }
            case (WaitForReregistration) {
              fl_IMS_DefaultHandler_Reregistration(v_DefaultHandlerInfo.IpcanType, v_IMS_DATA_REQ, v_NonceCountForNextRegister);  // @sic R5-163201 sic@
              v_NonceCountForNextRegister := v_NonceCountForNextRegister + 1;
            }
            case (WaitForDeregistration) {
              f_IMS_PTC_ResetRanType();                 // @sic R5s150057 change 3: reset RanType to allow de-registrtion from any RAT sic@
              f_IMS_PTC_Deregistration(v_IMS_DATA_REQ); // @sic R5-132037 sic@
              f_IMS_TCP_CloseHandling(waitAndClose);    /* @sic R5s170445: closing of TCP connection (after R5s140219 moving it into f_IMS_PTC_Deregistration) sic@ */
              f_IMS_GlobalRelease();                    /* the UE is not registered anymore => reset IMS
                                                         * @sic R5s150231 change 2 sic@
                                                         * reset IMS_PTC and IP_PTC's registration info in case of de-registration or the UE being switched off (-> cmr_IPCAN_IMS_Reset)
                                                         * NOTE: f_IMS_TCP_CloseHandling and f_IMS_GlobalRelease both are closing the TCP connections (if any);
                                                         *       nevertheless f_IMS_TCP_CloseHandling cannot be removed as additionally it has a delay before closing the connections
                                                         *       => even though f_IMS_TCP_CloseHandling(justWait) should do the same, there is no need to change this (@sic R5s170580 sic@) */
              v_DefaultHandlerInfo.State := WaitForRegistration;
            }
            case (NoRegistration) {                     // @sic R5-169150 sic@
              f_SetVerdictInconc(__FILE__, __LINE__, "IMS registration not allowed (e.g. as UE is in test loop mode)");
            }
            case else {
              FatalError(__FILE__, __LINE__, "IMS DefaultHandler: invalid state");
            }
          }
        }
      }
    }
  }




After Change:
	  function f_IMS_DefaultHandler() runs on IMS_PTC
  {
    var IMS_DefaultHandlerInfo_Type v_DefaultHandlerInfo;       /* @sic R5-163201 sic@ */
    var IMS_DATA_REQ v_IMS_DATA_REQ;
    var IMS_IPCAN_Coordination_MSG v_IMS_IPCAN_Coordination_MSG;
    var integer v_NonceCountForNextRegister;

    f_IMS_PTC_Init(-, defaultHandling, IMS_Security);
    f_IMS_PTC_CheckAudioVideoTags_Set(false);                   /* @sic R5-201110: set check of audio/video feature tag to be false sic@ */

    v_DefaultHandlerInfo.State := WaitForRegistration;
    v_DefaultHandlerInfo.IpcanType := IPCAN_E;                  /* @sic R5s160133 sic@ */

    while (true) {
      alt {
        [] IPCAN.receive(cmr_IPCAN_IMS_Control) -> value v_IMS_IPCAN_Coordination_MSG {         /* @sic R5s160133 sic@ */
          v_DefaultHandlerInfo := fl_IMS_DefaultHandler_ChangeInfo(v_DefaultHandlerInfo, IPCAN_E, v_IMS_IPCAN_Coordination_MSG.IPCAN_IMS_Control);
        }
        [] OtherIPCAN.receive(cmr_IPCAN_IMS_Control) -> value v_IMS_IPCAN_Coordination_MSG {    /* @sic R5s160133 sic@ */
          v_DefaultHandlerInfo := fl_IMS_DefaultHandler_ChangeInfo(v_DefaultHandlerInfo, OtherIPCAN_E, v_IMS_IPCAN_Coordination_MSG.IPCAN_IMS_Control);
        }
        [] OtherIPCAN.receive(cmr_IMS_IPCAN_Trigger (?)){}
        [] IMS_Server.receive(car_IMS_Register_Request(?)) -> value v_IMS_DATA_REQ {      // receive any REGISTER
          select (v_DefaultHandlerInfo.State) {
            case (WaitForRegistration) {
              f_IMS_DefaultHandler_InitialRegistration(v_DefaultHandlerInfo.IpcanType, v_IMS_DATA_REQ);  // @sic R5s140788 change 2 sic@
              v_DefaultHandlerInfo.State := WaitForDeregistration;
              v_NonceCountForNextRegister := 2;  /* next (re-)registration shall have nounce count of 2 @sic R5-163201 sic@ */
            }
            case (WaitForReregistration) {
              fl_IMS_DefaultHandler_Reregistration(v_DefaultHandlerInfo.IpcanType, v_IMS_DATA_REQ, v_NonceCountForNextRegister);  // @sic R5-163201 sic@
              v_NonceCountForNextRegister := v_NonceCountForNextRegister + 1;
            }
            case (WaitForDeregistration) {
              f_IMS_PTC_ResetRanType();                 // @sic R5s150057 change 3: reset RanType to allow de-registrtion from any RAT sic@
              f_IMS_PTC_Deregistration(v_IMS_DATA_REQ); // @sic R5-132037 sic@
              f_IMS_TCP_CloseHandling(waitAndClose);    /* @sic R5s170445: closing of TCP connection (after R5s140219 moving it into f_IMS_PTC_Deregistration) sic@ */
              f_IMS_GlobalRelease();                    /* the UE is not registered anymore => reset IMS
                                                         * @sic R5s150231 change 2 sic@
                                                         * reset IMS_PTC and IP_PTC's registration info in case of de-registration or the UE being switched off (-> cmr_IPCAN_IMS_Reset)
                                                         * NOTE: f_IMS_TCP_CloseHandling and f_IMS_GlobalRelease both are closing the TCP connections (if any);
                                                         *       nevertheless f_IMS_TCP_CloseHandling cannot be removed as additionally it has a delay before closing the connections
                                                         *       => even though f_IMS_TCP_CloseHandling(justWait) should do the same, there is no need to change this (@sic R5s170580 sic@) */
              v_DefaultHandlerInfo.State := WaitForRegistration;
            }
            case (NoRegistration) {                     // @sic R5-169150 sic@
              f_SetVerdictInconc(__FILE__, __LINE__, "IMS registration not allowed (e.g. as UE is in test loop mode)");
            }
            case else {
              FatalError(__FILE__, __LINE__, "IMS DefaultHandler: invalid state");
            }
          }
        }
      }
    }
  }
  



 
